
Come utilizzare questo documento
Questo documento è rivolto principalmente ai provider di servizi Internet (ISP) per la creazione di 
procedure che gli utenti finali potranno modificare per le proprie connessioni. 

Utilizzo delle procedure in Internet Explorer
IEscript.exe consente a un utente finale di utilizzare un file di procedura, ovvero un file con 
estensione .scp, per effettuare una connessione. IESCRIPT aggiunge le impostazioni al file di 
connessione, con estensione .con, nel modo descritto di seguito. Utilizzare IESCRIPT per 
modificare il file di connessione. Non modificare il file di connessione manualmente.

Nome sezione: Script

Sono supportate le seguenti parole chiave:

ScriptEnabled
Yes/No  Specifica se attivare l'utilità di creazione delle procedure. 

ScriptFileName
Una stringa completa che specifica la posizione del file di procedura associato con il file di 
connessione.

ScriptTerminal
Yes/No  Specifica se la finestra del terminale per la creazione delle procedure deve essere 
caricata durante la riproduzione della procedura.

ScriptRecord
Yes/No  Specifica se l'eseguibile della procedura deve essere in modalità di registrazione 
all'attivazione.

Esempio di valori del file di connessione per la creazione di procedure
Esempio 1:

[Script]
ScriptEnabled=Yes
ScriptFileName=C:\Shiva\Proced.Scp
ScriptTerminal=Yes
ScriptRecord=No

Esempio 2:

[Script]
ScriptEnabled=Yes
ScriptFileName=C:\Shiva\Nuproced.Scp
ScriptTerminal=Yes
ScriptRecord=Yes



Nota
· L'Esempio 2 illustra ciò che viene visualizzato mentre una procedura è in fase di 
creazione.

Creazione di una procedura in Internet Explorer
Per creare un file di procedura per un utente finale, è necessario in primo luogo creare il file di 
procedura con Scripter.exe. Dopo aver creato tale file, l'utente finale potrà modificarlo e inserirvi il
proprio nome e la propria password.

Creazione e riproduzione delle procedure con la finestra TTY
1 Eseguire Scripter.exe, che si trova in genere nella stessa cartella del programma di 

Internet Explorer. 
2 Scegliere Registra dal menu File.
3 Individuare un file di procedura, ovvero un file con estensione .scp, selezionarlo 

dall'elenco dei file. In alternativa, immettere un nuovo nome di file. Quando si avvia 
la registrazione, verrà visualizzata la finestra seguente.

4 Aprire il gruppo di programmi Internet Explorer e scegliere Nuova connessione. Seguire
le istruzioni visualizzate per creare una nuova connessione oppure scegliere l'icona di 
una connessione già effettuata. 

5 Selezionare l'icona della connessione. Scegliere Proprietà dal menu File. Assicurarsi di 
aver selezionato la casella di controllo Visualizza finestra del terminale dopo la 
connessione.

6 Completare la connessione al server Internet.
Nella finestra dell'applicazione per i file di procedure ShivaRemote verrà visualizzata la 
voce Started!!!.

7 Fare clic nella finestra di registrazione, che corrisponde correntemente alla finestra del 
terminale, e immettere le informazioni necessarie per la connessione al server ISP.

8 Dopo aver stabilito la connessione, scegliere Finisci registrazione nella finestra 
dell'applicazione per i file di procedure.

9 Nella finestra dell'applicazione ShivaPPP per i file di procedure scegliere OK.
10 Per chiudere la comunicazione, scegliere Annulla nella finestra di dialogo della 

connessione.

Nota  
Quando si utilizza questo file di procedura, assicurarsi che la casella di controllo Visualizza 
finestra del terminale dopo la connessione sia deselezionata.

Dopo avere immesso le informazioni richieste nella procedura, scegliere il pulsante Termina. La 
finestra verrà chiusa e avrà inizio la normale negoziazione PPP.

Di seguito viene descritta la procedura creata da questo programma:

; C:\SCRIPTS\IE\LOGIN.SCP
; Creato il:  7-1-1996 alle 17:07:11
;
;
proc main
string szPassword
transmit "^M"
waitfor "Nome Host:  ",matchcase until 15
transmit "spry01^M"
waitfor "UIC: ",matchcase until 6
transmit "spry053514^M"



waitfor "Password: ",matchcase until 3
if $PASSWORD then
 transmit $PASSWORD
 transmit "^M"
 goto doneTxPassword
endif
getinput "Password: " szPassword
transmit szPassword
transmit "^M"
doneTxPassword:
endproc

Modificare un file di procedura basato sul file registrato o creare un nuovo 
file di procedura
Scripter.exe registra un file di procedura contenente informazioni specifiche sull'utente, incluso il 
nome utente e la password. I provider di servizi Internet potranno fornire un file di procedura 
generico che potrà essere modificato e utilizzato da più utenti finali.

Utilizzare la seguente procedura di esempio come guida per la creazione di una procedura 
generica che non contenga informazioni specifiche sull'utente.

;---------------------------------------------------
; Nome file: LOGIN01.SCP
; Data Creazione: 18-10-1996
; Ora  Creazione: 12:00:00
;---------------------------------------------------
proc main
; Alcuni sistemi richiedono la pressione di un tasto
; per avviare la procedura di connessione
;--------------------------------------------------------------
transmit "^M"

; Invia il nome dell'host, fornito dal provider
; di servizi
;---------------------------------------------------
waitfor "Host Name:"
transmit "hostname"
transmit "^M"

; Invia il nome utente digitato nella casella
; Nome utente del file di connessione associato
;---------------------------------------------------------
waitfor "Username:"
transmit $USERID
transmit "^M"

; Invia la password digitata nella casella Password
; del file di connessione associato



;--------------------------------------------------------
waitfor "Password:"
transmit $PASSWORD
transmit "^M"
endproc

Informazioni sulla creazione di procedure avanzate

1.0   Cenni preliminari
Molti provider di servizi Internet e di servizi in linea richiedono l'inserimento manuale delle 
informazioni, quali nome utente e password, per stabilire la connessione. Con il supporto per la 
creazione di procedure di accesso remoto, è possibile scrivere una procedura per automatizzare 
tale processo. 

Una procedura è un file di testo che contiene una serie di comandi, parametri ed espressioni 
richiesti dal provider di servizi Internet o di servizi in linea per stabilire la connessione e utilizzare i
servizi. È possibile utilizzare un qualunque editor di testo, quale il Blocco note, per creare file di 
procedura. Dopo aver creato il proprio file di procedura, sarà possibile assegnarlo ad una 
specifica procedura per la connessione di accesso remoto utilizzando l'Utilità per la creazione di 
procedure per l'accesso remoto.

2.0   Struttura di base in una procedura
Un comando è l'istruzione di base contenuta in un file di procedura. Alcuni comandi richiedono 
parametri per definire ulteriormente la funzione del comando. Un'espressione è una 
combinazione di operatori e argomenti che generano un risultato. Le espressioni possono essere 
utilizzate come valori in un comando. Esempi di espressioni sono le operazioni aritmetiche, gli 
operatori relazionali e la concatenazione di stringhe.

La forma base di una procedura di accesso remoto è la seguente:

;
; Un commento comincia con un punto e virgola (;) e
; prosegue nella riga successiva.
;

proc main
;  Una procedura può essere costituita da un numero qualunque di 
;  variabili e comandi

dichiarazione variabili

blocco comandi

endproc

Una procedura deve essere costituita da una procedura principale, specificata dalla parola chiave
proc, e da una corrispondente parola chiave endproc che indica la fine della procedura.  

Prima di inserire i comandi è necessario dichiarare le variabili. Il primo comando viene eseguito 
nella procedura principale e ciascun comando successivo viene eseguito nell'ordine in cui appare
nella procedura. La procedura termina quando si raggiunge la fine della procedura principale.



3.0   Variabili
Le procedure possono contenere delle variabili. I nomi di variabile devono cominciare con una 
lettera o con un segno di sottolineatura ( _ ) e possono contenere una sequenza qualsiasi di 
lettere maiuscole o minuscole, numeri e segni di sottolineatura. Non è possibile utilizzare parole 
riservate come nomi di variabile. Per ulteriori informazioni, vedere l'elenco di parole riservate alla 
fine di questo documento.

Le variabili devono essere dichiarate prima di essere utilizzate. Quando si dichiara una variabile, 
è necessario definirne anche il tipo. Una variabile di un determinato tipo può contenere solo valori
dello stesso tipo. I tre tipi di variabile supportati i seguenti:

Tipo Integer
Descrizione Un numero positivo o negativo, quale 7, -12 o 5698.

Tipo String
Descrizione Una serie di caratteri racchiusi tra virgolette doppie, ad 
esempio "Ciao mondo!" o "Immettere la password:".

Tipo Boolean
Descrizione Un valore logico booleano quale TRUE o FALSE.

I valori vengono assegnati alle variabili utilizzando la seguente istruzione di assegnazione:

variabile = espressione

La variabile assume il valore dell'espressione valutata.  

Esempi:

integer count = 5
integer timeout = (4 * 3)
integer i

boolean bDone = FALSE

string szIP = (getip 2)

set ipaddr szIP

L'esempio precedente utilizza comandi di procedure non supportati da Shiva.

3.1   Variabili di sistema
Le variabili di sistema sono impostate da comandi di procedure o sono determinate dalle 
informazioni che vengono inserite quando si stabilisce una connessione. Le variabili di sistema 
sono di sola lettura, il che significa che non possono essere modificate all'interno della procedura.
Le variabili di sistema sono:

Nome $USERID
Tipo String

Descrizione L'identificazione dell'utente per la connessione corrente. 
Questa variabile è il valore del nome utente specificato nella 
finestra di dialogo Connessione a del programma di connessione.

Nome $PASSWORD 



Tipo String
Descrizione La password per la connessione corrente. Questa 
variabile è il valore del nome utente specificato nella finestra di dialogo 
Connessione a del programma di connessione.

Nome $SUCCESS
Tipo Boolean

Descrizione Questa variabile è impostata da alcuni comandi per 
indicare se il comando è riuscito o meno. Una procedura può assumere 
decisioni in base al valore di questa variabile.

Nome $FAILURE
Tipo Boolean

Descrizione Questa variabile è impostata da alcuni comandi per 
indicare se il comando è riuscito o meno. Una procedura può assumere 
decisioni in base al valore di questa variabile.

Queste variabili possono essere utilizzate in corrispondenza di un'espressione di tipo simile. Ad 
esempio, 

transmit $USERID

è un comando valido perché $USERID è una variabile di tipo stringa.

4.0   Stringhe
Le procedure per l'accesso remoto supportano l'utilizzo di sequenze di escape e conversioni 
apice, come descritto di seguito:

Stringa ^char
Descrizione Conversione apice

Se char è un valore compreso tra '@' e '_', la sequenza di caratteri viene 
convertita in valori a singolo byte compresi tra 0 e 31. ^M viene ad 
esempio convertito in un ritorno a capo.

Se char è un valore compreso tra 'a' e 'z', la sequenza di caratteri viene 
convertita in valori a singolo byte compresi tra 1 e 26.

Se char è un qualunque altro valore, la sequenza di caratteri non viene 
trattata in maniera particolare.

Stringa <cr>
Descrizione Ritorno a capo 

Stringa <lf>
Descrizione Avanzamento riga

Stringa \"
Descrizione Virgolette doppie

Stringa \^
Descrizione Accento circonflesso singolo

Stringa \<
Descrizione Segno di minore di (<) singolo



Stringa \\
Descrizione Barra rovesciata

Esempi:

transmit "^M"
transmit "Joe^M"
transmit "<cr><lf>"
waitfor "<cr><lf>"

5.0  Espressioni
Un'espressione è una combinazione di operatori e argomenti che vengono valutati per fornire un 
risultato. Le espressioni possono essere utilizzate come valori in un qualunque comando.

Un'espressione può combinare ciascuna variabile, ciascun valore intero, stringa o booleano con 
ciascun operatore unario e binario della tabella seguente. Tutti gli operatori unari hanno priorità 
più alta. La priorità degli operatori binari è indicata dalla corrispondente posizione nella tabella. 

Gli operatori unari sono i seguenti:

Operatore -
Tipo di operazione Meno unario

Operatore !
Tipo di operazione Complemento

Gli operatori binari sono elencati nella tabella seguente in ordine di priorità. Gli operatori con 
priorità più alta sono elencati per primi:

Operatori *  /
Tipo di operazione Moltiplicativo
Restrizioni di tipo Integer

Operatori +  -
Tipo di operazione Additivo per gli interi
Restrizioni di tipo String (solo +)

Operatori < > <= >=
Tipo di operazione Relazionale
Restrizioni di tipo Integer

Operatori == !=
Tipo di operazione Uguaglianza
Restrizioni di tipo Integer, string, boolean

Operatori and
Tipo di operazione AND logico
Restrizioni di tipo Boolean

Operatori or
Tipo di operazione OR logico
Restrizioni di tipo Boolean

Esempi:



count = 3 + 5 * 40
transmit "Ciao" + " laggiù"
delay 24 / (7 - 1)

6.0   Commenti
Tutto il testo su una riga successivo a un punto e virgola (;) viene ignorato.

Esempi:

; questo è un commento

transmit "ciao" ; trasmette la stringa "ciao"

7.0   Parole chiave
Le parole chiave specificano la struttura della procedura. Al contrario dei comandi, non eseguono 
azioni. Le parole chiave sono elencate di seguito.

proc nome 

Indica l'inizio di una procedura. Tutte le procedure devono avere una procedura principale (proc 
main). L'esecuzione della procedura comincia dalla procedura principale e termina alla fine di 
questa.

endproc

Indica la fine di una procedura. Quando si esegue una procedura e si arriva all'istruzione endproc
della procedura principale, l'applicazione per la creazione delle procedure di accesso remoto 
avvierà PPP o SLIP. 

integer nome [ = valore ] 

Dichiara una variabile di tipo intero. È possibile utilizzare una qualunque espressione numerica o 
una variabile per inizializzare la variabile.

string nome [ = valore ] 

Dichiara una variabile di tipo string. È possibile utilizzare una qualunque stringa o variabile di tipo 
stringa per inizializzare la variabile.

boolean nome [ = valore ] 

Dichiara una variabile di tipo boolean. È possibile utilizzare una qualunque espressione booleana
per inizializzare la variabile.

8.0   Comandi
L'implementazione iniziale di un linguaggio per la creazione di procedure per la Connessione 
telefonica Internet Explorer sarà un sottoinsieme del linguaggio per la creazione di procedure per 
l'accesso remoto specificato per Microsoft Windows(R) 95.  

Tutti i comandi sono parole riservate, il che significa che non è possibile dichiarare variabili che 
hanno lo stesso nome dei comandi. I comandi sono i seguenti:

delay nSecondi

Attende per il numero di secondi specificato da nSecondi prima di eseguire il successivo 
comando della procedura.



Esempio:

delay 2 ; attende per 2 secondi
delay x * 3 ; attende per x * 3 secondi

goto etichetta

Salta alla riga della procedura specificata da etichetta e continua ad eseguire i comandi 
successivi a questa riga.

Esempio:

 waitfor "Prompt>" until 10
 if !$SUCCESS then

goto BailOut ; salta a BailOut ed esegue il comando
; ad essa successivo

 endif

 transmit "bbs^M"
 goto End

BailOut:
 transmit "^M"

halt

Arresta la procedura. Questo comando non rimuove la finestra di dialogo del terminale. 
Per stabilire la connessione, è necessario scegliere il pulsante Continua. Non è possibile
riavviare la procedura.

if condizione then
comandi

endif

Esegue la serie di comandi comandi se la condizione condizione ha valore TRUE.

Esempio:

if $USERID == "John" then
transmit "Johnny^M"

endif

etichetta :

Specifica la riga della procedura alla quale saltare. Un'etichetta deve essere un nome 
unico e seguire la convenzione per l'assegnazione dei nomi alle variabili.

transmit stringa [ , raw ]

Invia i caratteri specificati da stringa al computer remoto. 

Il computer remoto riconoscerà le sequenze di escape e le conversioni apice a meno che



non si includa il parametro raw nel comando. Il parametro raw è utile quando si 
trasmettono variabili di sistema quali $USERID e $PASSWORD nel caso in cui il nome 
utente o la password contengano caratteri che, senza il parametro raw, sarebbero 
interpretate come sequenze di escape o conversioni apice.

Esempi:

transmit "slip" + "^M"
transmit $USERID, raw

waitfor stringa [ , matchcase ] [ then etichetta 
{ , stringa [ , matchcase ] then etichetta } ]
[ until tempo ]

Attende che il computer riceva dal computer remoto una o più stringhe specificate. Il 
parametro stringa non rileva la distinzione tra maiuscole e minuscole a meno che non si 
inserisca il parametro matchcase.

Se si riceve una stringa richiesta ed è stato utilizzato il parametro then etichetta, questo 
comando passerà alla riga del file della procedura individuata da etichetta.

Il parametro opzionale until tempo definisce il massimo numero di secondi che il 
computer dovrà attendere per ricevere la stringa prima di eseguire il comando 
successivo. Senza questo parametro, il computer attenderà per un tempo indefinito.

Se il computer riceve una delle stringhe specificate, la variabile di sistema $SUCCESS è 
impostata a TRUE, altrimenti è impostata a FALSE se il numero di secondi specificati da 
tempo trascorre prima che sia stata ricevuta la stringa.  

Esempio:

waitfor "Accesso:"

waitfor "Password?", matchcase

waitfor "prompt>" until 10

waitfor
"Accesso:" then DoLogin,
"Password:" then DoPassword,
"BBS:" then DoBBS,
"Altro:" then DoOther

 until 10

Il comando seguente è un'estensione Shiva all'insieme di comandi per le procedure definito da 
Microsoft.

getinput "Mostra stringa " szToReceiveInput

Indica all'utente di inserire un'informazione durante l'esecuzione di una procedura.

Esempio:

Getinput "Password di sistema" szSystemPassword  



9.0   Parole riservate
Le parole seguenti sono riservate e non possono essere utilizzate come nome di variabile:

and boolean databits delay
do endif endproc endwhile
even FALSE getip goto
halt if integer ipaddr
keyboard mark matchcase none
odd off on or
parity port proc raw
screen set space stopbits
string then transmit TRUE
until waitfor while

10.0 Comandi Shiva non supportati dall'applicazione per la creazione 
delle procedure
getip
port databits
port parity
port stopbits
set screen keyboard
ipaddr
while/endwhile


